A Spectral Decomposition

```
> restart;
  with(LinearAlgebra):
```

Example #1: We will consider a symmetric matrix A (so it must have a spectral decomposition).

```
> A := <<1,2,1>|<2,1,-1>|<1,-1,0>>;
```

$$A := \begin{bmatrix} 1 & 2 & 1 \\ 2 & 1 & -1 \\ 1 & -1 & 0 \end{bmatrix} \quad \textbf{(1)}$$

```
> Lambda,P := Eigenvectors(A);
```

$$\Lambda, P := \begin{bmatrix} 1 \\ 3 \\ -2 \end{bmatrix}, \begin{bmatrix} \frac{1}{2} & 1 & -1 \\ -\frac{1}{2} & 1 & 1 \\ 1 & 0 & 1 \end{bmatrix} \quad \textbf{(2)}$$

Since A has 3 distinct eigenvalues (so that its eigenspaces are 1 dimensional), we just need to normalize these eigenvectors to get an orthonormal basis for 3-space.

In general, we would need to run Gram-Schmidt on each eigenspace's basis (and normalize) to get our orthonormal basis.

Q = [v1 v2 v3] is then an orthogonal matrix (i.e., its inverse is its transpose),
and A can be orthgonally diagonalized by Q.

```
> w1 := P.<1,0,0>; v1 := w1/sqrt(w1.w1);
  w2 := P.<0,1,0>; v2 := w2/sqrt(w2.w2);
  w3 := P.<0,0,1>; v3 := w3/sqrt(w3.w3);

  Q := <v1|v2|v3>;

  'Q^T*Q' = Transpose(Q).Q;

  'Q^T*A*Q' = Transpose(Q).A.Q;
```

$$w1 := \begin{bmatrix} \frac{1}{2} \\ -\frac{1}{2} \\ 1 \end{bmatrix}$$

$$v1 := \begin{bmatrix} \frac{\sqrt{6}}{6} \\ -\frac{\sqrt{6}}{6} \\ \frac{\sqrt{6}}{3} \end{bmatrix}$$

$$w2 := \begin{bmatrix} 1 \\ 1 \\ 0 \end{bmatrix}$$

$$v2 := \begin{bmatrix} \frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} \\ 0 \end{bmatrix}$$

$$w3 := \begin{bmatrix} -1 \\ 1 \\ 1 \end{bmatrix}$$

$$v3 := \begin{bmatrix} -\frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \end{bmatrix}$$

$$Q := \begin{bmatrix} \frac{\sqrt{6}}{6} & \frac{\sqrt{2}}{2} & -\frac{\sqrt{3}}{3} \\ -\frac{\sqrt{6}}{6} & \frac{\sqrt{2}}{2} & \frac{\sqrt{3}}{3} \\ \frac{\sqrt{6}}{3} & 0 & \frac{\sqrt{3}}{3} \end{bmatrix}$$

$$Q^T Q = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{3}$$

$$Q^T A Q = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & -2 \end{bmatrix} \quad (3)$$

Now for the spectral decomposition. Here we write A as a sum of projection matrices scaled by eigenvalues.

The projection matrices are built from our orthnormal basis. These matrices T1, T2, and T3 are orthogonal to each other (i.e., their products are zero) and they sum up to be the identity matrix.

Each of these scaled by its corresponding eigenvalue then summed up yields the spectral decomposition of A.

Note that the transpose of v_i times a vector v yields the i-th coordinate of v (since the v_i's form an orthonormal basis). Thus v_i Transpose(v_i) v is nothing more than the (orthogonal projection) of v onto v_i. Thus T_i = v_i Transpose(v_i) yield a matrix that collapses a vector onto the subspace spanned by v_i.

For example, if an eigenspace had an orthonormal basis {u, v, w}, the corresponding projection matrix would be T = u Transpose(u) + v Transpose(v) + w Transpose(w).

Once our projection matrices are built, say T_1, ..., T_k corresponding to eigenspaces for lambda_1, ..., lambda_k, then T_1 + ... + T_k = I and our spectral decomposition is A = lambda_1 T_1 + ... + lambda_k T_k.

```
> T1 := v1.Transpose(v1);
  T2 := v2.Transpose(v2);
  T3 := v3.Transpose(v3);

  'T[1] + T[2] + T[3]' = T1+T2+T3;

  'lambda[1]*T[1]+lambda[2]*T[2]+lambda[3]*T[3]' = Lambda[1]*T1+Lambda
  [2]*T2+Lambda[3]*T3;
```

$$T1 := \begin{bmatrix} \frac{1}{6} & -\frac{1}{6} & \frac{1}{3} \\ -\frac{1}{6} & \frac{1}{6} & -\frac{1}{3} \\ \frac{1}{3} & -\frac{1}{3} & \frac{2}{3} \end{bmatrix}$$

$$T2 := \begin{bmatrix} \frac{1}{2} & \frac{1}{2} & 0 \\ \frac{1}{2} & \frac{1}{2} & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

$$T3 := \begin{bmatrix} \frac{1}{3} & -\frac{1}{3} & -\frac{1}{3} \\ -\frac{1}{3} & \frac{1}{3} & \frac{1}{3} \\ -\frac{1}{3} & \frac{1}{3} & \frac{1}{3} \end{bmatrix}$$

$$T_1 + T_2 + T_3 = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$T_1 \lambda_1 + T_2 \lambda_2 + T_3 \lambda_3 = \begin{bmatrix} 1 & 2 & 1 \\ 2 & 1 & -1 \\ 1 & -1 & 0 \end{bmatrix} \quad (4)$$

Note projection matrices are in fact projection matrices (i.e., equal to their own squares), and our projections are orthogonal to each other (i.e., products are zero between distinct matrices).

```
> T1 = T1^2;
  T1.T2;
```

$$\begin{bmatrix} \frac{1}{6} & -\frac{1}{6} & \frac{1}{3} \\ -\frac{1}{6} & \frac{1}{6} & -\frac{1}{3} \\ \frac{1}{3} & -\frac{1}{3} & \frac{2}{3} \end{bmatrix} = \begin{bmatrix} \frac{1}{6} & -\frac{1}{6} & \frac{1}{3} \\ -\frac{1}{6} & \frac{1}{6} & -\frac{1}{3} \\ \frac{1}{3} & -\frac{1}{3} & \frac{2}{3} \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad (5)$$

Example #2: This matrix is lifted from problem #4c) on Fall 2016 Test #3.

We calculate eigenvalues and eigenvectors. Then we need to run Gram-Schmidt on each eigenspace basis (so this step can be skipped for any non-repeated eigenvalue).

At this point we have an orthogonal basis for eigenvectors, so we normalize them and get an orthonormal basis of eigenvectors. Q (built from this basis) is an orthogonal matrix that diagonalizes A.

```
> A := <<1,3,3>|<3,1,3>|<3,3,1>>;

  Lambda, P := Eigenvectors(A);

  # Maple lists eigenvalues in different orders according to its
  # randomized factorization algorithm. So we hard code the following
```

```
# solution so this always works...
Lambda := <-2,-2,7>;
w1 := <-1,0,1>; # instead of P.<1,0,0>;
w2 := <-1,1,0>; # instead of P.<0,1,0>;
w3 := <1,1,1>;  # instead of P.<0,1,0>;

# Gram-Schmidt run on the -2 eigenspace basis:
u1 := w1;
u2 := w2-(w2.u1)/(u1.u1)*u1;

u3 := w3;

# Normalize to get an orthonormal basis:
v1 := u1/sqrt(u1.u1);
v2 := u2/sqrt(u2.u2);
v3 := u3/sqrt(u3.u3);

Q := <v1|v2|v3>;

'Q^T*Q' = Transpose(Q).Q;

'Q^T*A*Q' = Transpose(Q).A.Q;
```

$$A := \begin{bmatrix} 1 & 3 & 3 \\ 3 & 1 & 3 \\ 3 & 3 & 1 \end{bmatrix}$$

$$\Lambda, P := \begin{bmatrix} 7 \\ -2 \\ -2 \end{bmatrix}, \begin{bmatrix} 1 & -1 & -1 \\ 1 & 0 & 1 \\ 1 & 1 & 0 \end{bmatrix}$$

$$\Lambda := \begin{bmatrix} -2 \\ -2 \\ 7 \end{bmatrix}$$

$$w1 := \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}$$

$$w2 := \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$$

$$w3 := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$$

$$u1 := \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}$$

$$u2 := \begin{bmatrix} -\frac{1}{2} \\ 1 \\ -\frac{1}{2} \end{bmatrix}$$

$$u3 := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$$

$$v1 := \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ 0 \\ \frac{\sqrt{2}}{2} \end{bmatrix}$$

$$v2 := \begin{bmatrix} -\frac{\sqrt{6}}{6} \\ \frac{\sqrt{6}}{3} \\ -\frac{\sqrt{6}}{6} \end{bmatrix}$$

$$v3 := \begin{bmatrix} \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \end{bmatrix}$$

$$Q := \begin{bmatrix} -\frac{\sqrt{2}}{2} & -\frac{\sqrt{6}}{6} & \frac{\sqrt{3}}{3} \\ 0 & \frac{\sqrt{6}}{3} & \frac{\sqrt{3}}{3} \\ \frac{\sqrt{2}}{2} & -\frac{\sqrt{6}}{6} & \frac{\sqrt{3}}{3} \end{bmatrix}$$

$$Q^T Q = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$Q^T A Q = \begin{bmatrix} -2 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & 7 \end{bmatrix} \quad (6)$$

We have the following spectral decomposition:

```
> v1 := Q.<1,0,0>;
  v2 := Q.<0,1,0>;
  v3 := Q.<0,0,1>;

  # T1 projects onto the -2 eigenspace and
  # T2 projects onto the 7 eigenspace
  T1 := v1.Transpose(v1) +  v2.Transpose(v2);
  T2 := v3.Transpose(v3);

  'T[1]+T[2]' = T1+T2;

  '-2*T[1]+7*T[2]' = -2*T1+7*T2;
```

$$v1 := \begin{bmatrix} -\frac{\sqrt{2}}{2} \\ 0 \\ \frac{\sqrt{2}}{2} \end{bmatrix}$$

$$v2 := \begin{bmatrix} -\frac{\sqrt{6}}{6} \\ \frac{\sqrt{6}}{3} \\ -\frac{\sqrt{6}}{6} \end{bmatrix}$$

$$v3 := \begin{bmatrix} \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \end{bmatrix}$$

$$T1 := \begin{bmatrix} \frac{2}{3} & -\frac{1}{3} & -\frac{1}{3} \\ -\frac{1}{3} & \frac{2}{3} & -\frac{1}{3} \\ -\frac{1}{3} & -\frac{1}{3} & \frac{2}{3} \end{bmatrix}$$

$$T2 := \begin{bmatrix} \frac{1}{3} & \frac{1}{3} & \frac{1}{3} \\ \frac{1}{3} & \frac{1}{3} & \frac{1}{3} \\ \frac{1}{3} & \frac{1}{3} & \frac{1}{3} \end{bmatrix}$$

$$T_1 + T_2 = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$-2\ T_1 + 7\ T_2 = \begin{bmatrix} 1 & 3 & 3 \\ 3 & 1 & 3 \\ 3 & 3 & 1 \end{bmatrix} \tag{7}$$